
There are a number of techniques that
can be used within ISPF to accomplish
what is commonly referred to as a

point-and-shoot operation. Over the years, I
have dug around within control blocks and used
edit macro services to make data strings beneath
the cursor position available to my dialogs.

With the addition of a few simple "Z" vari-
ables to the standard ISPF environment, IBM
now provides the easiest way yet to accomplish
point-and-shoot operations. Because of this, I
am not going to get bogged down explaining
all the older techniques. I have abandoned
most of them anyway. Instead, I'll just forge
ahead with the basics of an ultra-simple
COBOL dialog function that can serve as the
basis of your own point-and-shoot routine.

WHY POINT-AND-SHOOT?

Laziness might be one answer but laziness
in a productive context. If there is a big long
complex dataset name on the screen, why
should an ISPF user have to type (or copy and
paste) it all over again? Why not just point to
the dataset on the screen with your mouse,
wherever you happen to be and then invoke a
browse, edit, view against it or obtain infor-
mation about it?

Typing avoidance is what's at play here. This
is always productive, especially when the names
to be typed are cryptic and long. PC environ-
ments have this mastered. Why not ISPF?

POINT-AND-SHOOT
ANYWHERE

Before I begin the explanation, I want to
make it clear that the routine explained this
month is capable of obtaining data at the cur-
rent cursor coordinates anywhere in ISPF.

Not just in Edit or View, but at any screen—
IBM or third-party, SDSF, IOF. But it is highly

dependant upon ISPF itself. Specifically, it
relies on the availability of the "ZSCREENx"
variables shown in FIGURE 1.

THE ZSCREEN VARIABLES

These IBM "Z" variables are critical to
accomplishing a clean and simple point-and-
shoot. Several of these variables have been
around for years but the two most important
ones first appeared in OS/390 V2R10. These
two new variables are named ZSCREENI and
ZSCREENC.

If you've read my columns over the years,
something I always recommend is keeping
your eyes peeled for new things that appear in
ISPF. These might be new services (like
QUERYENQ). They might also be new
parameters that can be used with older ISPF
services. Or, in the case of the ZSCREEN vari-
ables, they might be new variables that IBM
provides. Their true value might not be imme-
diately apparent. That's my job, explaining
how to exploit these new-fangled things.

ZSCREENI is a copy of the entire screen
image. Think of it as a large character field.
What is its length? That depends. I'll explain
shortly.

ZSCREENC is the relative displacement of
the cursor (minus one) into the large character
string named ZSCREENI.

The other two variables that are needed in a
point-and-shoot are ZSCREENW and
ZSCREEND. ZSCREENW stores the width of

the terminal screen (in bytes) and ZSCREEND
holds the depth (in rows). On a standard
"Model 2" terminal, these values would be 24
(ZSCREEND) and 80 (ZSCREENW).

This is how the length of ZSCREENI is
determined: Width times Depth (or 24 X 80).

WHERE IS THE CURSOR
POINTING?

The cursor-pointing value is contained in
the variable named ZSCREENC. It is stored as
an integer that represents a displacement into
the ZSCREENI variable. If the ZSCREENI
variable was of a standard 24 X 80 image
(1,920 bytes) and the ZSCREENC variable
contained a value of 162, this would indicate
that the cursor was sitting on top of the 163rd
byte of ZSCREENI.

Since the "hard displacement" of the cursor
position is provided in ZSCREENC all that
really needs to be done is to determine the row
and column of the cursor. And frankly, even
this need not be done. What is needed is the
remaining length of the row AFTER the posi-
tion where the cursor is pointing (or to the
"right" of the cursor).

For example, if the cursor is on top of the
163rd byte of a 24 X 80 image, it is really at
Row 2, Column 3 of the screen. Knowing the
column and the screen width (provided in
ZSCREENW), allows for a simple subtraction
of width (80) minus cursor column (3) to pro-
vide the remaining length (77).

You can look these up in the appropriate ISPF Reference Summary for whatever release of ISPF you are running.
ZSCREENI and ZSCREENC were not available until OS/390 V2R10.

ZSCREENW – Contains the width (in bytes) of the terminal
ZSCREEND – Contains the depth (in rows) of the terminal
ZSCREENI – Contains the current screen image as a character string
ASCREENC – Contains the cursor position as a displacement into ZSCREENI

FIGURE 1: THE ZSCREEN VARIABLES

CONCLUSION

The CLCCPASM routine in FIGURE 2 is
pretty bare bones. It uses a COBOL DISPLAY
for output. But the essentials of the point-and-
shoot are contained within it. Here are a few
things to know regarding execution:

� Allocate SYSOUT to your TSO session
by issuing: ALLOC DD(SYSOUT)
DA(*)

� Link-edit the COBOL program into an
ISPLLIB library, a TSOLIB tasklib or
some load library where implicit
execution can be done.

� Invoke the routine by entering:
Command→ TSO CLCCPASM from
any ISPF command line. Don't forget to
move the cursor around the screen to see
different results.

Feel free to enhance and otherwise integrate
the routine into your own environment. You
could, for example, parse the string of data
that the cursor is pointing to and look at it in
"pieces." You could also include sensitivity
variables for the cursor position—right on a
word, to the left of a word or even anywhere
on a row (line).

NaSPA member Jim Moore is the president of
Concentrated Logic Corporation, a Glendale
Heights, IL-based software development firm spe-
cializing in TSO/ISPF/PDF and database design. He
can be reached at conlogco@comcast.net.

Note that the COPY parameter of VDEFINE is used to pull in the shared pool values of the ZSCREENx variables as
part of their VDEFINE.

000100 Identification Division.
000200 Program-Id. Clccpasm.
000300 Author. Jim Moore - Concentrated Logic Incorporated.
000400*--Remarks.------------------------------------------------------- *
000500*                                                         *
000600* This Program Is An "ISPF Dialog Function" written in COBOL. It *
000700* uses the IBM-provided "ZSCREEN" variables to locate the exact *
000800* cursor co-ordinates. The text at the cursor, as well as the   *
000900* cursor "row and column" are displayed.                    *
001000*                                                         *
001100* Use it as a starting point for "Point and Shoot" operations *
001200* within your own ISPF dialogs. Hint: Return the text at the   *
001300* cursor - both parsed and in total - to the active Shared    *
001400* pool.                                                     *
001500*----------------------------------------------------------------- *
001600 Environment Division.
001700 Data Division.
001800 Working-Storage Section.
001900 01 Ispf-Api-Services.
002000   05  Vdefine         Pic  X(08) Value "VDEFINE".
002100   05  Vput                Pic  X(08) Value "VPUT".
002200   05  Setmsg              Pic  X(08) Value "SETMSG".
002300 01 Ispf-Api-Parameters.
002400  05  Fixd                Pic  X(08) Value "FIXED".
002500    05  Char                Pic  X(08) Value "CHAR".
002600   05  Copy-Parm           Pic  X(08) Value "COPY".
002700   05  Ispz000             Pic  X(08) Value "ISPZ000".
002800   05  Isplink             Pic  X(08) Value "ISPLINK".
002900    05  Len-Fld             Pic S9(08) Binary.
003000 01 Vdefine-Literals.
003100   05  Len4-Fixed         Pic  X(28)
003200         Value "(ZSCREENW,ZSCREEND,ZSCREENC)".
003300   05  Len-Var-Screen     Pic  X(10)
003400         Value "(ZSCREENI)".
003500 01 Function-Pool.
003600   05  Bin-Fld.
003700         10  Screen-Width   Pic S9(08) Binary.
003800         10  Screen-Depth   Pic S9(08) Binary.
003900         10  Cursor-Pos     Pic S9(08) Binary.
004000     05  Screen-Image       Pic  X(8192) Value Spaces.
004100 01  Working-Variables.
004200    05  Cursor-Row          Pic S9(08) Binary.
004300    05  Cursor-Column       Pic S9(08) Binary.
004400    05  Len-to-End-of-Row  Pic S9(08) Binary.
004500    05  Work-Field          Pic S9(03)V999.
004600    05                     Redefines Work-Field.
004700        10  Work-Integer    Pic  9(03).
004800        10  Work-Decimal    Pic SV999.
004900/
005000 Procedure Division.
005100     Move Length Of Screen-Width To Len-Fld.
005200     Call Isplink Using Vdefine Len4-Fixed Bin-Fld Fixd Len-Fld
005300                              Copy-Parm.
005400     Compute Len-Fld = Screen-Width * Screen-Depth.
005500     Call Isplink Using Vdefine Len-Var-Screen Screen-Image Char
005600                              Len-Fld Copy-Parm.
005700     Add 1 To Cursor-Pos.
005800
005900     Compute Work-Field = Cursor-Pos / Screen-Width.
006000
006100     If Work-Decimal > Zeros
006200        Compute Cursor-Column = Cursor-Pos -
006300            (Work-Integer * Screen-Width)
006400        Compute Len-to-End-of-Row =
006500            (Screen-Width - Cursor-Column)
006600        Compute Cursor-Row = Work-Integer + 1
006700     Else
006800     Move Screen-Width To Cursor-Column
006900     Move Zeros        To Len-to-End-of-Row
007000     Move Work-Integer To Cursor-Row.
007100
007200     Display "Cursor at Row " Cursor-Row " Column " Cursor-Column.
007300     Display "Pointing to-> "
007400             Screen-Image (Cursor-Pos:Len-to-End-of-Row).
007500     Goback.

FIGURE 2: THE CLCCPASM BASELINE POINT-AND-SHOOT ROUTINE


